Proceedings of the International Geoscience and Remote Sensing Symposium IGARSS'99, 28 June - 02 July1999, Hamburg, Germany, ed. Tammy I. Stein, IEEE Cat. No. 99CH36293, ISBN 0-7803-5207-6, IEEE, Piscataway, NJ, USA, 1999.

# Two-term Henyey-Greenstein light scattering phase function for seawater

Vladimir I. Haltrin

Naval Research Laboratory, Ocean Sciences Branch, Code 7331, Stennis Space Center, MS 39529-5004, USA
phone: 228-688-4528, fax: 228-688-5379, e-mail: <haltrin@nrlssc.navy.mil>

**Abstract** — In this presentation a one-parameter model of marine two-term Henyey-Greenstein (TTHG) phase function is proposed. The original three-parameter TTHG phase function was reduced to a one-parameter TTHG phase function by utilizing experimental regression dependencies between two integral parameters of the phase function, average cosine and average square of cosine of scattering angle, with the probability of backscattering. An algorithm and FORTRAN program to calculate one parameter two-term Henyey-Greenstein phase function of scattering is presented.

## INTRODUCTION

An analytic Henyey-Greenstein (HG) scattering phase function is very popular in radiative transfer calculations in astrophysics [1] and atmospheric and oceanic optics [2]. This phase function is very convenient for numeric and Monte Carlo calculations because it has a simple representation in the form of Legendre polynomials series and it allows easy calculation of asymptotic radiance distributions. Unfortunately, the shape of the HG phase function inadequately represents the shape of realistic atmospheric and marine phase functions. For the case of atmospheric optics this shortcoming was successfully resolved by G. Kattawar [3] who proposed a two-term Henyey-Greenstein phase function (TTHG). Unfortunately, Kattawar's TTHG phase function is not applicable to marine waters because it utilizes specific properties of atmospheric phase functions. In this paper a similar approach is adopted but for oceanic water.

## APPROACH

The Henyey-Greenstein phase function [3, 4] has the following analytic form:

$$p_{HG}(\mu, g) = \frac{1-g^2}{\left(1-2g\mu+g^2\right)^{3/2}}, \quad \mu = \cos\vartheta, \tag{1}$$

here $\vartheta$ is a scattering angle and parameter $g$ is equal to the average cosine $\overline{\cos\vartheta}$ over angular distribution (1) defined as:

$$\overline{\cos\vartheta} = \frac{1}{2}\int_{-1}^{1} p(\mu)\,\mu\,d\mu, \quad \frac{1}{2}\int_{-1}^{1} p(\mu)\,d\mu = 1. \tag{2}$$

Phase function (1) has the following representation in Legendre polinomial series:

$$p_{HG}(\mu, g) = \sum_{n=0}^{\infty}(2n+1)\,g^n P_n(\mu), \quad 0 \le g \le 1. \tag{3}$$

An average of squared cosine for phase function (1) is:

$$\overline{\cos^2\vartheta} = \frac{1}{2}\int_{-1}^{1} p_{HG}(\mu)\,\mu^2\,d\mu = \frac{1}{3}\left(1+2g^2\right), \tag{4}$$

and probability of backscattering is defined by equation [4]:

$$B = \frac{1}{2}\int_{-1}^{0} p_{HG}(\mu, g)\,d\mu = \frac{1-g}{2g}\left(\frac{1+g}{\sqrt{1+g^2}} - 1\right). \tag{5}$$

Henyey-Greenstein phase function has remarkable analytical properties, but it is ill-suited to model scattering in real seawater because its shape differs significantly from the shape of marine phase functions [4].

To overcome this shortcomings and retain all desirable properties of HG phase function a TTHG phase function was chosen as proposed by Kattawar [3]:

$$p_{TTHG}(\mu, \alpha, g, h) = \alpha\, p_{HG}(\mu, g) + (1-\alpha)\, p_{HG}(\mu, -h), \quad 0 \le \alpha, g, h \le 1. \tag{6}$$

The phase function (6) has the following decomposition in Legendre polinomial series:

$$p_{TTHG}(\mu, \alpha, g, h) = \sum_{n=0}^{\infty}(2n+1)\left[\alpha\, g^n + (1-\alpha)(-h)^n\right]P_n(\mu). \tag{7}$$

Its integral parameters are given by the following formulae:

$$B = \alpha\frac{1-g}{2g}\left(\frac{1+g}{\sqrt{1+g^2}} - 1\right) + (1-\alpha)\frac{1+h}{2h}\left(1 - \frac{1-h}{\sqrt{1+h^2}}\right), \tag{8}$$

$$\overline{\cos\vartheta} = \alpha(g+h) - h, \tag{9}$$

$$\overline{\cos^2\vartheta} = \frac{1}{3} + \frac{2}{3}\left[\alpha\left(g^2 - h^2\right) + h^2\right]. \tag{10}$$

In order to reduce the number of parameters $(\alpha, g, h)$ in this phase function (6) a relationship between integral parameters is derived from experimental measurements. A number of regression relationships between integral parameters of phase function are given in paper by Timofeyeva [5]. These relationships are rewritten in form convenient for the purposes to eliminate the extra parameters in Eq. (6). These new regressions are represented as follows:

$$\overline{\cos\vartheta} = 2\frac{1-2B}{2+B}, \quad 0.05 \le B \le 0.25, \tag{11}$$

$$\overline{\cos^2\vartheta} = \frac{6-7B}{3(2+B)}, \quad 0.05 \le B \le 0.25. \tag{12}$$

Relationships (11)-(12) are considered much better than the original ones given in Ref. [5] because they give values of parameters that lie in the range of experimental error for $0.05 \le B \le 0.25$ and satisfy asymptotic conditions at $B = 0$ (delta-shaped scattering) and $B = 0.5$ (isotropic scattering).

By solving relationships (11),(12) with Eqs. (8)-(10), we have the following connections between parameters $(\alpha, h)$ and parameter $g$:




Figure 1. Dependence of integral parameters $(B, \overline{\cos\vartheta}, \overline{\cos^2\vartheta})$ of the marine TTHG phase function on parameter g in Eq. (7).




Figure 2. Samples of marine TTHG phase functions for different values of parameter $g$ computed with the program `TTHGmwater`.

$$\alpha = \frac{h(1+h)}{(g+h)(1+h-g)}, \tag{13}$$

$$h = -0.3061446 + 1.000568\,g - 0.01826332\,g^2 + 0.03643748\,g^3, \qquad 0.30664 \le g \le 1. \tag{14}$$

Substitution of Eqs. (13) and (14) in Eq. (7) gives us a one-parameter two-term Henyey-Greenstein phase function of scattering with integral parameters $(B, \overline{\cos\vartheta}, \overline{\cos^2\vartheta})$ adjusted to the experimental relationships given by Eqs. (11) and (12).

## CONCLUSION

It is shown that a realistic marine phase function may be represented as one parameter two-term Henyey-Greensteyn phase function that has the same dependencies between integral characteristics of the phase function. In some cases this phase function may be more preferable than Fournier-Forand [6] or analytic form of Petzold phase functions [4]. The proposed phase function may be very convenient for radiative transfer modeling [7-10].

## ACKNOWLEDGMENT

The author thanks continuing support at the Naval Research Laboratory through the Littoral Optical Environment (LOE 6640-09) program. This article represents NRL contribution NRL/PP/7331–99–0003.

## APPENDIX: A PROGRAM TO CALCULATE ONE PARAMETER MARINE TWO-TERM HENYEY-GREENSTEIN SCATTERING PHASE FUNCTION

```
c            program TTHGmwater
c    written by Vladimir I. Haltrin, NRL Code 7331
c       e-mail: <haltrin@nrlssc.navy.mil>
c       This program is free for public use
        implicit    none
        integer     i
        real*10     g,gmin,eps,B,mu,m2
        real*10     ang(361),phf(361)
        character   tb
        data        gmin /0.30664/,eps /0.000001/

        open(11, file='g.in', status='old')
           read(11,*) g
        close(11)

        if (g.lt.gmin) g = gmin
        if (g.gt.(1.-eps)) g = 1.-eps

        call calcall(g, B,mu,m2)
        call calcphf(g, ang,phf)

        tb = CHAR(9)
        open(22, file='tthgphf.out', status='new')
           write(22,77) 'g = ', g
           write(22,77) 'B = ',B
           write(22,77) '<mu> = ',mu
           write(22,77) '<mu^2> = ',m2
           write(22,*)
           write(22,88) 'ang',tb,'phf'
```

```
      do i=1, 361
            write(22,99) ang(i),tb,phf(i)
      end do
   close(22)

77 format(a9,g12.5)
88 format(a3,a1,a3)
99 format(f6.1,a1,g12.5)
   end

   subroutine calcall(g, B,mu,m2)
   implicit   none
   real*10    g,B,mu,m2, falfa,fb, h,alf,h2
   h = -0.3061446+g*(1.000568+g*(0.03643748*g
  &    -0.01826332))
   alf = falfa(g,h)
   mu  = alf*(g+h)-h
   h2  = h*h
   h2  = alf*(g*g-h2)+h2
   h2  = h2+h2
   m2  = (1.+h2)/3.
   B = alf*fb(g)+(1.-alf)*fb(-h)
   return
   end

   subroutine calcphf(g, ang,phf)
   implicit   none
   integer    i
   real*10    g,B,mu,m2, falfa,fb,fpHG
   real*10    h,alf,h2, angi
   real*10    ang(361),phf(361)
   h = -0.3061446+g*(1.000568+g*(0.03643748*g
  &    -0.01826332))
   alf = falfa(g,h)
   do i=1,361
      angi = 0.5*(i-1)
      ang(i) = angi
      mu = COSD(angi)
      phf(i) = alf*fpHG(mu,g)+(1.-alf)
  &            *fpHG(mu,-h)
   end do
   return
   end

   real*10 function fpHG(mu,g)
   implicit   none
   real*10    mu,g,g2,z
   g2 = g*g
   z = mu*g
   z = 1.-z-z+g2
   z = z*SQRT(z)
   fpHG = (1.-g2)/z
   return
   end

   real*10 function falfa(g,h)
   implicit   none
   real*10     g,h,z
   if (g.eq.0.0.or.g.eq.1.0) then
      falfa = 1.
   else
      z = (g+h)*(1.-g+h)
      falfa = h*(1.+h)/z
   end if
   return
   end

   real*10 function fb(g)
   implicit   none
   real*10     g,f
   if (g.eq.0.) then
      f = 1.0
   else
      f = (1.-1./g)*(1.-(1.+g)/SQRT(1.+g*g))
   end if
   fb = 0.5*f
   return
   end
```

```
-------------------------------------------------
file "g.in":
0.9995        <-- g  (0.30664 ≤ g ≤ 1.0)
```

## REFERENCES


[1] H. C. Van de Hulst, *Multiple Light Scattering*, **I**, (Academic Press, New York - San Francisco, 1980).

[2} C. D. Mobley, *Light and Water* (Academic Press, San Diego–Toronto, 1994).

[3] G. W. Kattawar, "A three-parameter analytic phase function for multiple scattering calculations," *J. Quant. Spectrosc. Radiat. Transfer*, **15**, pp. 839-849, (1975).

[4] V. I. Haltrin, "Theoretical and empirical phase functions for Monte Carlo calculations of light scattering in seawater," in *Proceedings of the Fourth International Conference Remote Sensing for Marine and Coastal Environments*, **I**, pp. 509-518, (Envir. Res. Inst. of Michigan, Ann Arbor, MI, 1997).

[5] V. A. Timofeyeva, "Relation between Light-Field Parameters and between Scattering Phase Function Characteristics of Turbid Media, Including Seawater," *Izv. Atmos. Ocean Physics*, **14**, pp. 843-848, (1978).

[6] V. I. Haltrin, "An analytic Fournier-Forand scattering phase function as an alternative to the Henyey-Greenstein phase function in hydrologic optics," in *1998 IEEE International Geoscience and Remote Sensing Symposium Proceedings*, **II**, T. I. Stein, ed., (IEEE, Piscataway, NJ, USA, 1998), 910-912.

[7] V. I. Haltrin, "Self-consistent approach to the solution of the light transfer problem for irradiances in marine waters with arbitrary turbidity, depth and surface illumination," *Appl. Optics*, **37**, 3773-3784 (1998).

[8] V. I. Haltrin, "Apparent optical properties of the sea illuminated by Sun and sky," *Appl. Optics*, **37**, 8336-8340 (1998).

[9] V. I. Haltrin, "Diffuse reflection coefficient of a stratified sea," *Appl. Optics*, **38**, 932-936 (1999).

[10] V. I. Haltrin (*a.k.a.* В. И. Халтурин), "Propagation of Light in the Sea Depth," in *Optical Remote Sensing of the Sea and the Influence of the Atmosphere*, V. A. Urdenko and G. Zimmermann, eds., pp. 20-62 (German Democratic Republic Academy of Sciences Institute for Space Research, Berlin, 1985).